Raabo.

A/S REGNECENTRALEN
SCANDINAVIAN INFORMATION PROCESSING SYSTEMS

SYSTEM LIBRARY

RCSL NO: 53-M8
EDITION: September 1970
AUTHOR : Bo Jacoby

beta

KEYWORDS: RC 4000, Software, beta, Algol Procedure, ISO Tape

ABSTRACTS: beta(x, y) approximates the beta function.
beta(x, y) = integral from 0 till 1 of(1-t)××(x-1)×t××(y-1)×dt

INFORMATION DEPARTMENT
DK-2500 VALBY · BJERREGAARDSVEJ 5 · PHONE: (01) 46 08 88 · TELEX: 6464 rcinf dk · CABLES: INFOCENTRALEN

Beta function,beta(x,y)

1. Function and parameters.

beta(x,y) approximates the beta function.

beta(x,y)= integral from 0 till 1 of(1-t)××(x-1)×t××(y-1)×dt

procedure heading:

```
real procedure beta(x,y);
value x,y; real x,y;
```

procedure identifier:

beta : (real)
approximate function of arguments not resulting in under - or overflow, in which case beta is undefined.

call parameters:

x,y : (real or integer)
arguments.

2. Method.

The value of beta(x,y) is calculated in the range 1<=x<=2, 1<=y<=2 by means of the formula

beta(x,y)=gamma(x+1)×gamma(y+1)/x/y/gamma(x+y) or
beta(x,y)=gamma(x+1)×gamma(y+1)/x/y/(x+y-1)/gamma(x+y-1)
according to whether x+y<=3 or x+y>3.

The value of gamma(z) is approximated in the range 2<=z<=3 by a rational function of z-2, which is given as approximation 5231 in reference (1).

For arguments outside the range 1<=x<=2, 1<=y<=2, reductions are performed according to the formula:

beta(x+1,y)=x/(x+y)×beta(x,y)

3. Accuracy and time requirement.

The maximum relative error will be about

$$max(1,(abs(x)+abs(y))\times_{10}-10$$

The c.p.u.-time used for a call of beta is crudely

$$5+0.1\times(abs(x)+abs(y)) \text{ milliseconds.}$$

4. Test.

testprogram and output:

```
begin
 real b,x,y;
 for overflows:=0 while read(in,x,y)=2
 do
  begin
   b:=beta(x,y);
   write(out,<:<10>x=:>,<<-ddd.d>,x,<:   y=:>,
         y,<: beta(x,y)=:>,<<dddddddddddd000₁₀-ddd>,b,
         <<    dd₁₀-dd>,abs(b-gamma(x)×gamma(y)/gamma(x+y))          );
   setposition(out,0,0);
  end;
end;
x=   0.5  y=   0.5 beta(x,y)=  314159265376010 -12     010  0
x=   1.0  y=   1.0 beta(x,y)=  100000000000010-12    5810-12
x= 100.0  y=   1.0 beta(x,y)=100000000004000₁₀-16    45₁₀-14
x=  10.0  y=  -0.5 beta(x,y)= -107833813244001₀-12     0₁₀  0
```

5. Algol procedure.

```
beta=set 2
beta=algol
external
real procedure beta(x,y);
value x,y;
real x,y;

begin
 real h,w;
 for w:=0,x
 do
 begin
  if w=0
  then h:=1
  else
  begin
   x:=y;
   y:=w
  end
  ;
  if x>2
  then
  begin
   for x:=x-1 step -1 until 1
   do h:=h×x/(x+y);
   x:=x+1
  end
  else
  if x<1
  then
  for x:=x step 1 until 1
  do h:=h×(x+y)/x
 end
 ;
 w:=x+y-1;
 if w>2
 then
```

```
 begin
  h:=h/(w×x×y);
  w:=w-2
 end
 else
 begin
  h:=h/(x×y);
  w:=w-1
 end
 ;
 for w:=((((((
          .0393013l46419 ×w
         +.142928007949)×w
        +1.09850630453 )×w
        +3.36954359131 )×w
       +12.8021698112  )×w
       +22.9680800836  )×w
       +43.9410209189  )
              /
         (((             w
         -7.15075063299)×w
         +4.39050474596)×w
        +43.9410209191 )
  while y>0
 do
 begin
  if x>0
  then
  begin
   h:=h/w;
   w:=x-1;
   x:=0
  end
  else
  begin
   h:=h×w;
   w:=y-1;
   y:=0
  end
 end
```

```
 ;
 beta:=h×w
end beta;
```

6. Reference.

(1) J.F. Hart and oth.:
Computer Approximations,
John Wiley and Sons, 1968, p. 130-136